Lossy Dielectric: matching two regions

PHYS 4020 3.0

February 28, 2013

Explanation

We define a material through its constituent relation constants.
σ is the conductivity
μ is the permeability
ϵ is the permittivity
We will find out by experimenting with the notebook that in the end the frequency of the wave
$\omega = 2\pi f$
plays an important role in determining the properties (resistive vs capacitive).

A dielectric with some conductivity is considered lossy, since a conduction current density will play a role that competes with the displacement current density; conduction in a medium leads to Joule heating, which implies a loss of power (which means that the electric and magnetic field amplitudes have to suffer from attenuation).
If the conductivity $\sigma = 0$, we have no losses, since there is no conduction, $j = \sigma E$ vanishes.
In an ideal conductor $\sigma = \infty$, and presumably the capacitive properties disappear, i.e., the displacement current density becomes negligible.

The parameter that determines the ratio of conduction to displacement current is
$\tan(\theta) = \sigma/(\omega\epsilon)$
which demonstrates the frequency dependence.

At large frequency conduction effects are suppressed.

In this notebook we consider two adjacent homogeneous regions: a wave approaches from the left, some of it is transmitted, and some is reflected. The regions are labeled as 1 and 2.

We define the basic parameters first:

In[5]:=
```
Clear["Global`*"]
ω1=N[2Pi 10^6]; (* 1 MHz *)
ϵ0=8.854 10^(-12);
μ0=N[4Pi 10^(-7)];
```

Now we choose our example parameter values: a case where conduction and displacement compete.
For region 1 we make the losses small (mostly dielectric medium):

In[9]:=
```
ω = 10 ω1; (* 10 MHz *)
ϵ1 = 14 ϵ0; (* relative permittivity = 14 *)
μ1 = 1.0 μ0; (* relative permeability = 1.1 *)
σ1 = 10^-4; (* conductivity in SI: *)
σ1 / ω / ϵ1
```
Out[13]=
```
0.0128396
```

For region 2 we choose a more conductive property:

In[14]:=
```
ϵ2 = 1.0 ϵ0; (* relative permittivity = 14 *)
μ2 = 1.1 μ0; (* relative permeability = 1.1 *)
σ2 = 10^-2; (* conductivity in SI: *)
σ2 / ω / ϵ2
```
Out[17]=
```
17.9755
```

In[18]:=
```
α1 = ω Sqrt[μ1 ϵ1 / 2 (Sqrt[1 + (σ1 / ω / ϵ1)^2] - 1)]
β1 = ω Sqrt[μ1 ϵ1 / 2 (Sqrt[1 + (σ1 / ω / ϵ1)^2] + 1)]
```
Out[18]=
```
0.00503422
```
Out[19]=
```
0.784201
```

In[20]:=
```
α2 = ω Sqrt[μ2 ϵ2 / 2 (Sqrt[1 + (σ2 / ω / ϵ2)^2] - 1)]
β2 = ω Sqrt[μ2 ϵ2 / 2 (Sqrt[1 + (σ2 / ω / ϵ2)^2] + 1)]
(* β = k is the wavenumber, while α gives the attenuation *)
```
Out[20]=
```
0.640918
```
Out[21]=
```
0.677564
```

Observe how α and β acquired almost the same value, since $\sigma/\omega/\epsilon$ is large.

In[22]:=
```
η1 = Sqrt[I ω μ1 / (σ1 + I ω ϵ1)]
(* complex impedance *)
```
Out[22]=
```
100.68 + 0.646322 i
```

In[23]:=
```
η2 = Sqrt[I ω μ2 / (σ2 + I ω ϵ2)]
```
Out[23]=
```
67.6518 + 63.9928 i
```

In[24]:=
```
{Abs[η1], Abs[η2]}
(* controls the strength of the H field compared to E,
the imaginary part of η -> the phase by which H lags behind E *)
```
Out[24]=
```
{100.682, 93.1227}
```

In[25]:=
```
E0i = 1; (* 1 Volt/meter *)
H0i = E0i / η1
```
Out[26]=
```
0.00993202 - 0.0000637591 i
```

In LossyDielectric2.PDF we derive the matching conditions, and the resulting reflection and transmission coefficients (which are complex amplitudes, since the impedances $\eta 1$ and $\eta 2$ are complex).

In[27]:=
```
rf = (η2 - η1) / (η1 + η2)
```
Out[27]=
```
-0.0450604 + 0.393622 i
```

In[28]:=
```
tr = 2 η2 / (η1 + η2)
```
Out[28]=
```
0.95494 + 0.393622 i
```

In[29]:=
```
Efield1 = ComplexExpand[
  Re[E0i (Exp[-α1 z] Exp[I (ω t - β1 z)] + rf Exp[α1 z] Exp[I (ω t + β1 z)])]]
```
Out[29]=

$$e^{0.-0.00503422\,z}\,\mathrm{Cos}[0.+6.28319\times10^{7}\,t-0.784201\,z]-0.0450604\,e^{0.+0.00503422\,z}\,\mathrm{Cos}[0.+6.28319\times10^{7}\,t+0.784201\,z]-0.393622\,e^{0.+0.00503422\,z}\,\mathrm{Sin}[0.+6.28319\times10^{7}\,t+0.784201\,z]$$

In[30]:=
```
Hfield1 = ComplexExpand[
  Re[H0i ( Exp[-α1 z] Exp[I (ω t - β1 z)] - rf Exp[α1 z] Exp[I (ω t + β1 z)])]]
```
Out[30]=

$$0.00993202\,e^{0.-0.00503422\,z}\,\mathrm{Cos}[0.+6.28319\times10^{7}\,t-0.784201\,z]+0.000422444\,e^{0.+0.00503422\,z}\,\mathrm{Cos}[0.+6.28319\times10^{7}\,t+0.784201\,z]+0.0000637591\,e^{0.-0.00503422\,z}\,\mathrm{Sin}[0.+6.28319\times10^{7}\,t-0.784201\,z]+0.00391233\,e^{0.+0.00503422\,z}\,\mathrm{Sin}[0.+6.28319\times10^{7}\,t+0.784201\,z]$$

In[31]:=
```
T = 2 Pi / ω
λ1 = 2 Pi / β1
λ2 = 2 Pi / β2
```
Out[31]=

$1.\times10^{-7}$

Out[32]=
```
8.01221
```
Out[33]=
```
9.2732
```

In order to show the electric and magnetic field strength on the same graph we boost the magnetic strength by 75% of its attenuation factor.

In[34]:=

```
PL1 = Plot[{Efield1 /. t → T / 2, 0.75 Abs[η1] Hfield1 /. t → T / 4},
  {z, -2 λ1, 0}, PlotRange → All, PlotStyle → {Blue, Green}]
```

Out[34]=



In[35]:=

```
Efield2 = ComplexExpand[Re[E0i tr Exp[-α2 z] Exp[I (ω t - β2 z)]]]
```

Out[35]=

$$0.95494\, e^{0.-0.640918\, z} \operatorname{Cos}\left[0.+6.28319\times 10^{7}\, t-0.677564\, z\right]-0.393622\, e^{0.-0.640918\, z} \operatorname{Sin}\left[0.+6.28319\times 10^{7}\, t-0.677564\, z\right]$$

In[36]:=

```
Hfield2 = ComplexExpand[Re[E0i tr / η2 Exp[-α2 z] Exp[I (ω t - β2 z)]]]
```

Out[36]=

$$0.0103545\, e^{0.-0.640918\, z} \operatorname{Cos}\left[0.+6.28319\times 10^{7}\, t-0.677564\, z\right]+0.00397609\, e^{0.-0.640918\, z} \operatorname{Sin}\left[0.+6.28319\times 10^{7}\, t-0.677564\, z\right]$$

In[37]:=

```
PL2 = Plot[{Efield2 /. t → T / 2, 0.75 Abs[η1] Hfield2 /. t → T / 4},
  {z, 0, λ2}, PlotRange → All, PlotStyle → {Blue, Green}]
```

Out[37]=



In[38]:=

```
Show[{PL1, PL2}, PlotRange → All]
```

Out[38]=



To visualize the wave nature we animate the plot over time.

In[39]:=

```
Animate[Show[
  Plot[Evaluate@{Efield1 /. t → s, 0.75 Abs[η1] Hfield1 /. t → s} , {z, -2 λ1, 0},
   PlotRange → {{-2 λ1, λ2}, {-2 E0i, 2 E0i}}, PlotStyle → {Blue, Green}],
  Plot[Evaluate@{Efield2 /. t → s, 0.75 Abs[η1] Hfield2 /. t → s} , {z, 0, λ2},
   PlotRange → {{-2 λ1, λ2}, {-2 E0i, 2 E0i}}, PlotStyle → {Blue, Green}]],
 {s, 0, 2 T}, AnimationRunning → False]
```

Out[39]=



In the limit that region 2 represents a perfect conductor the fields in region 1 are standing waves:
E1_x = 2E0i sin(β1 z) sin(ωt),   and
H1_y = (2E0i/η1) cos(β1 z) cos(ωt) .

The standing waves are phase-shifted by 90 degrees, and the temporal oscillations are also out of phase by the same amount.
Note how the magnetic field has antinodes at the metal surface.

Inside the lossy region the phase delay of H with respect to E is maximally 45 degrees.